Logic

An Idris port of Coq.Init.Logic

Eric Bailey
https://github.com/yurrriq
eric@ericb.me

ABSTRACT

Here I present an Idris port of the `Coq.Init.Logic` module from the Coq standard library.

Keywords

logic, coq, idris

```
||| An Idris port of Coq.Init.Logic
module Logic

import Data.Bifunctor

%access export
```

1. PROPOSITIONAL CONNECTIVES

1.1 Unit

`()` is the always true proposition ($\top$).

```
%elim data Unit = MkUnit
```

1.2 Void

`Void` is the always false proposition ($\bot$).

```
%elim data Void : Type where
```

1.3 Negation

`Not a`, written `~a`, is the negation of `a`.

```
syntax "~" [x] = (Not x)
```

```
Not : Type -> Type
Not a = a -> Void
```

1.4 Conjunction

`And a b`, written `(a, b)`, is the conjunction of `a` and `b`.

`Conj p q` is a proof of `(a, b)` as soon as `p` is a proof of `a` and `q` a proof of `b`.

`proj1` and `proj2` are first and second projections of a conjunction.

```
syntax "(" [a] "," [b] ")" = (And a b)

||| The conjunction of `a` and `b`.
data And : Type -> Type -> Type where
     Conj : a -> b -> (a, b)

implementation Bifunctor And where
    bimap f g (Conj a b) = Conj (f a) (g b)

||| First projection of a conjunction.
proj1 : (a, b) -> a
proj1 (Conj a _) = a

||| Second projection of a conjunction.
proj2 : (a, b) -> b
proj2 (Conj _ b) = b
```

1.5 Disjunction

`Either a b` is the disjunction of `a` and `b`.

```
data Either : Type -> Type -> Type where
     Left   : a -> Either a b
     Right  : b -> Either a b
```

1.6 Biconditional

Proof Wiki

$$\dfrac{\varphi \vdash \psi \qquad \psi \vdash \varphi}{\vdash \varphi \iff \psi}$$

`iff a b`, written `a <-> b`, expresses the equivalence of `a` and `b`.

```
infixl 9 <->
```

```
/// The biconditional is a *binary connective* that
/// can be voiced: *p* **if and only if** *q*.
public export
(<->) : Type -> Type -> Type
(<->) a b = (a -> b, b -> a)
```

1.6.1 *Biconditional is Reflexive*

Proof Wiki

$$\dfrac{\varphi \vdash \varphi}{\vdash \varphi \iff \varphi}\,\mathcal{I}$$

```
/// The biconditional operator is reflexive.
iffRefl : a <-> a
iffRefl = Conj id id
```

1.6.2 *Biconditional is Transitive*

Proof Wiki

$$\dfrac{\dfrac{\wedge\mathcal{E}_1\,\dfrac{(\varphi \iff \psi) \wedge (\psi \iff \chi)}{\varphi \iff \psi \qquad \psi \iff \chi}\,\wedge\mathcal{E}_2}{\varphi \iff \chi}}{((\varphi \iff \psi) \wedge (\psi \iff \chi)) \implies (\varphi \iff \chi)}\,\mathcal{I}$$

```
/// The biconditional operator is transitive.
iffTrans : (a <-> b) -> (b <-> c) -> (a <-> c)
iffTrans (Conj ab ba) (Conj bc cb) =
    Conj (bc . ab) (ba . cb)
```

1.6.3 *Biconditional is Commutative*

Proof Wiki

$$\dfrac{\dfrac{\dfrac{\dfrac{\varphi \iff \psi}{(\varphi \implies \psi) \wedge (\psi \implies \varphi)}}{(\psi \implies \varphi) \wedge (\varphi \implies \psi)}}{\psi \iff \varphi} \qquad \dfrac{\dfrac{\dfrac{\psi \iff \varphi}{(\psi \implies \varphi) \wedge (\varphi \implies \psi)}}{(\varphi \implies \psi) \wedge (\psi \implies \varphi)}}{\varphi \implies \psi}}{\varphi \iff \psi \dashv\vdash \psi \iff \varphi}$$

or

$$\dfrac{\dfrac{\dfrac{\varphi \iff \psi}{\psi \iff \varphi}}{(\varphi \iff \psi) \implies (\psi \iff \varphi)} \implies \mathcal{I} \qquad \dfrac{\dfrac{\varphi \iff \psi}{\psi \iff \varphi}}{(\varphi \iff \psi) \implies (\psi \iff \varphi)} \implies \mathcal{I}}{(\varphi \iff \psi) \iff (\psi \iff \varphi)}$$

Source.

```
/// The biconditional operator is commutative.
iffSym : (a <-> b) -> (b <-> a)
iffSym (Conj ab ba) = Conj ba ab
```

1.6.4 *andIffCompatLeft*

$\psi \iff \chi \dashv\vdash (\varphi \wedge \psi) \iff (\varphi \wedge \chi)$

$$\wedge\mathcal{E}_1\,\dfrac{\Gamma, \varphi \vdash \psi \iff \chi}{\Gamma, \varphi \implies \psi \vdash \chi \quad \Gamma, \varphi \vdash \chi \implies \psi}\,\wedge\mathcal{E}_2$$

```
andIffCompatLeft : (b <-> c) -> ((a, b) <-> (a, c))
andIffCompatLeft = bimap second second
```

1.6.5 *andIffCompatRight*

$\psi \iff \chi \dashv\vdash (\psi \wedge \varphi) \iff (\chi \wedge \varphi)$

```
andIffCompatRight : (b <-> c) -> ((b, a) <-> (c, a))
andIffCompatRight = bimap first first
```

1.6.6 *orIffCompatLeft*

$\psi \iff \chi \vdash (\varphi \vee \psi) \iff (\varphi \vee \chi)$

```
orIffCompatLeft : (b <-> c) ->
                  (Either a b <-> Either a c)
orIffCompatLeft = bimap second second
```

1.6.7 *orIffCompatRight*

$\psi \iff \chi \vdash (\psi \vee \varphi) \iff (\chi \vee \varphi)$

```
orIffCompatRight : (b <-> c) ->
                   (Either b a <-> Either c a)
orIffCompatRight = bimap first first
```

1.6.8 *negVoid*

$\neg\varphi \dashv\vdash \varphi \iff \bot$

or

$\vdash \neg\varphi \iff (\varphi \iff \bot)$

```
negVoid : (~a) <-> (a <-> Void)
negVoid = Conj (flip Conj void) proj1
```

1.6.9 *andCancelLeft*

$$\dfrac{\psi \implies \varphi \quad \chi \implies \varphi}{((\varphi \wedge \psi) \iff (\varphi \wedge \chi)) \iff (\psi \iff \chi)}$$

```
andCancelLeft : (b -> a) ->
                (c -> a) ->
                (((a, b) <-> (a, c)) <-> (b <-> c))
andCancelLeft ba ca = Conj (bimap f g) andIffCompatLeft
  where
    f h b = proj2 . h $ Conj (ba b) b
    g h c = proj2 . h $ Conj (ca c) c
```

1.6.10 andCancelRight

```
andCancelRight : (b -> a) ->
                 (c -> a) ->
                 (((b, a) <-> (c, a)) <-> (b <-> c))
andCancelRight ba ca = Conj (bimap f g) andIffCompatRight
  where
    f h b = proj1 . h $ Conj b (ba b)
    g h c = proj1 . h $ Conj c (ca c)
```

1.6.11 Conjunction is Commutative

Proof Wiki

Formulation 1. $\varphi \wedge \psi \dashv\vdash \psi \wedge \varphi$

Formulation 2. $\vdash (\varphi \wedge \psi) \iff (\psi \wedge \varphi)$

Source.

```
||| Conjunction is commutative.
andComm : (a, b) <-> (b, a)
andComm = Conj swap swap
  where
    swap : (p, q) -> (q, p)
    swap (Conj p q) = Conj q p
```

1.6.12 Conjunction is Associative

Proof Wiki

Formulation 1. $(\varphi \wedge \psi) \wedge \chi \dashv\vdash \varphi \wedge (\psi \wedge \chi)$

Formulation 2. $\vdash ((\varphi \wedge \psi) \wedge \chi) \iff (\varphi \wedge (\psi \wedge \chi))$

Source.

```
||| Conjunction is associative.
andAssoc : ((a, b), c) <-> (a, (b, c))
andAssoc = Conj f g
  where
    f abc@(Conj (Conj a b) c) =
        Conj a (first proj2 abc)
    g abc@(Conj a (Conj b c)) =
        Conj (second proj1 abc) c
```

1.6.13 orCancelLeft

$(\psi \implies \neg\varphi) \implies (\chi \implies \neg\varphi) \implies (((\varphi \vee \psi) \iff (\varphi \vee \chi)) \iff (\psi \iff \chi))$

```
orCancelLeft : (b -> ~a) ->
               (c -> ~a) ->
               ((Either a b <-> Either a c) <->
                (b <-> c))
orCancelLeft bNotA cNotA =
    Conj (bimap f g) orIffCompatLeft
  where
    f ef b = go (bNotA b) (ef (Right b))
    g eg c = go (cNotA c) (eg (Right c))
    go : (~a) -> Either a b -> b
    go lf = either (void . lf) id
```

1.6.14 orCancelRight

$$\frac{\psi \vdash \neg\varphi \qquad \chi \vdash \neg\varphi}{((\psi \vee \varphi) \iff (\chi \vee \varphi)) \iff (\psi \iff \chi)}$$

```
orCancelRight : (b -> ~a) ->
                (c -> ~a) ->
                ((Either b a <-> Either c a) <->
                 (b <-> c))
orCancelRight bNotA cNotA =
    Conj (bimap f g) orIffCompatRight
  where
    f ef b = go (bNotA b) (ef (Left b))
    g eg c = go (cNotA c) (eg (Left c))
    go : (~p) -> Either q p -> q
    go rf = either id (void . rf)
```

1.6.15 Disjunction is Commutative

Proof Wiki

$(\varphi \vee \psi) \iff (\psi \vee \varphi)$

```
||| Disjunction is commutative.
orComm : Either a b <-> Either b a
orComm = Conj mirror mirror
```

1.6.16 Disjunction is Associative

Proof Wiki

$(\varphi \vee \psi) \vee \chi \vdash \varphi \vee (\psi \vee \chi)$

```
||| Disjunction is associative on the left.
orAssocLeft : Either (Either a b) c ->
              Either a (Either b c)
orAssocLeft = either (second Left) (pure . pure)
```

$\varphi \vee (\psi \vee \chi) \vdash (\varphi \vee \psi) \vee \chi$

```
||| Disjunction is associative on the right.
orAssocRight : Either a (Either b c) ->
               Either (Either a b) c
orAssocRight = either (Left . Left) (first Right)
```

Formulation 1. $(\varphi \vee \psi) \vee \chi \dashv\vdash \varphi \vee (\psi \vee \chi)$

Formulation 2. $\vdash ((\varphi \vee \psi) \vee \chi) \iff (\varphi \vee (\psi \vee \chi))$

Source.

```
||| Disjunction is associative.
orAssoc : Either (Either a b) c <->
          Either a (Either b c)
orAssoc = Conj orAssocLeft orAssocRight
```

1.6.17 iffAnd

$\varphi \iff \psi \vdash (\varphi \implies \psi) \wedge (\psi \implies \varphi)$

```
iffAnd : (a <-> b) -> (a -> b, b -> a)
iffAnd = id
```

1.6.18 iffAndTo

$\varphi \iff \psi \dashv\vdash (\varphi \implies \psi) \wedge (\psi \implies \varphi)$

or

$\vdash (\varphi \iff \psi) \iff ((\varphi \implies \psi) \wedge (\psi \implies \varphi))$

```
iffToAnd : (a <-> b) <-> (a -> b, b -> a)
iffToAnd = Conj id id
```